Newsgroup: it.comp.lang.visual-basic
Titolo: Dialog ctrl come winamp ?qual'e L'Api ?

Ciao a tutti, qualcuno sa risolvermi un problemino che mi sta facendo
perdere ore di sonno ?
Ho usato in un form VisualBasic il controllo Commondialog, e dopo aver perso
qualche minuto mi sono reso conto che con esso non è possibile aprire una
finestra di dialogo capace di visualizzare solo le directory.
Lo so che per questo c'è il controllo DirListBox ma non è la stessa cosa in
termini di bellezza/professionalita', e poi ho bisogno di usare il
CommonDialog anche perchè in un altro punto del mio prog. lo uso per
scegliere un file, e quindi vorrei usarne uno simile per le directory.
Chi conosce Winamp avra notato che per addizionare un file.mp3 il prog. usa
un controllo identico in tutto a CommonDialog di VB(in pratica comdlg32.ocx)
ma per Add. un 'intera DIR la finestra di dialogo è diversa (Forse lo stesso
ctrl  ma con una prorietà style settata divesamente ) Ovviamente il ctrl
CommonDialog di VB non Espone la proprietà Style.
Ora vengo al dunque:
Io ho provato con l' utility M$ Spy++ a vedere le differenze tra la finestra
CommonDialog del mio prog, e quella di winamp (quando si chede di add. un
file) e Sono IDENTICHE, mentre la finestra CommonDialog del mio prog, e
quella del add. directory di winamp sono diverse in questi punti:
 (WINAMP)
Class name     :   #32770 (Dialog)
Window styles:   14C820C4
(Mia app.)
Class name     :   #32770 (Dialog)
Window styles:   96CC20C4
C' è una chiamata API che mi permette di cambiare lo style della mia
finestra per farla diventare come quella di winamp ?? Io ho provato con
FindWindow ma ho solo attenuto dei GPF :-(
Grazie mille per avermi seguito fin qua ( duemila per chi mi da una mano
-)  )
Ciao a tutti.
Nicola.


Non so se è proprio quello che cerchi, ma questo apre una finestra di
dialogo per la scelta delle directory. Copia il seguente codice in un modulo
e poi chiama SelezionaPercorso.
Es: SelezionaPercorso(frmTest, "TitoloFinestra")
Option Explicit
Private Const BIF_RETURNONLYFSDIRS = 1
Private Const BIF_DONTGOBELOWDOMAIN = 2
Private Const MAX_PATH = 260
Private Declare Function SHBrowseForFolder Lib "shell32" _
                                  (lpbi As BrowseInfo) As Long
Private Declare Function SHGetPathFromIDList Lib "shell32" _
                                  (ByVal pidList As Long, _
                                  ByVal lpBuffer As String) As Long
Private Declare Function lstrcat Lib "kernel32" Alias "lstrcatA" _
                                  (ByVal lpString1 As String, ByVal _
                                  lpString2 As String) As Long
Private Type BrowseInfo
   hWndOwner      As Long
   pIDLRoot       As Long
   pszDisplayName As Long
   lpszTitle      As Long
   ulFlags        As Long
   lpfnCallback   As Long
   lParam         As Long
   iImage         As Long
End Type
Public Function SelezionaPercorso(ByRef F As Form, ByVal Titolo As String)
As String
'Opens a Treeview control that displays the directories in a computer
   Dim lpIDList As Long
   Dim sBuffer As String
   Dim tBrowseInfo As BrowseInfo
   With tBrowseInfo
      .hWndOwner = F.hWnd
      .lpszTitle = lstrcat(Titolo, "")
      .ulFlags = BIF_RETURNONLYFSDIRS + BIF_DONTGOBELOWDOMAIN
   End With
   lpIDList = SHBrowseForFolder(tBrowseInfo)
   If (lpIDList) Then
      sBuffer = Space(MAX_PATH)
      SHGetPathFromIDList lpIDList, sBuffer
      sBuffer = Left(sBuffer, InStr(sBuffer, vbNullChar) - 1)
      If Right(sBuffer, 1) <> "\" Then sBuffer = sBuffer & "\"
      SelezionaPercorso = sBuffer
   Else
      SelezionaPercorso = ""
   End If
End Function

 Non so se č proprio quello che cerchi, ma questo apre una finestra di


 
http://www.murialdo.it:8000/it_lang_vb/
 ma non va !! Bo...
Non c'e' piu' la porta:
http://www.murialdo.it/it_lang_vb/
Ciao
Aldo